



Tutorial Letter 201/1/2018

COMPUTER ALGEBRA

APM2616

Semester 1

Department of Mathematical Sciences

This tutorial letter contains solutions
for assignment 01.

BAR CODE

QUESTION 1

(a) Use the protocol command and WRITEPATH variable to output a portion of a MuPAD session to a file not in the current directory. **(5 Marks)**

(b) Use MuPAD to find the prime factors of an integer (Hint: see *ifactor*). **(5 Marks)**

[10 Marks]

SOLUTION 1

```
(a) WRITEPATH:="c:\\MuPAD\\2018\\";
```

```
"c:\\MuPAD\\2018\\"
```

```
protocol("assignment1.out");
```

```
a:=5;b:=8;a+b;
```

```
5
```

```
8
```

```
13
```

```
more mupad commands.....
```

```
protocol();
```

Open the file "assignment1.out" with wordpad to see the output.

(5 Marks)

```
(b) ifactor(1032);
```

```
      3  
     2 3 43
```

(5 Marks)
[10 Marks]

QUESTION 2

Use MuPAD to prove

$$\frac{2 \cos 5x}{\sin 2x \cos^2 x} = -10 \sin x + \frac{\cos^2 x}{\sin x} + \frac{5 \sin^3 x}{\cos^2 x}.$$

[10 Marks]

SOLUTION 2

```
a2:=simplify((2*cos(5*x)/(sin(2*x)*cos(x)^2)+10*sin(x)  
-cos(x)^2/sin(x)-5*sin(x)^3/cos(x)^2));
```

```
0
```

[10 Marks]**QUESTION 3**

Find the values of a, b and c for which the following matrix is not invertible $\begin{pmatrix} 1 & a & b \\ 1 & 1 & c \\ 1 & 1 & 1 \end{pmatrix}$.

[10 Marks]**SOLUTION 3**

```
export (linalg);
A:=matrix([[1,a,b],[1,1,c],[1,1,1]]);
```

```
+ - - +
| 1, a, b |
|         |
| 1, 1, c |
|         |
| 1, 1, 1 |
+ - - +
```

```
d:=det(A);
```

```
a c - c - a + 1
```

```
factor(d);
```

```
(a - 1) (c - 1)
```

Thus, the matrix is not invertible iff $a=1$ and/or $c=1$.

[10 Marks]**QUESTION 4**

Consider the following matrices $A = \begin{pmatrix} 1 & 3 & 0 \\ -1 & 2 & 7 \\ 0 & 8 & 1 \end{pmatrix}$, $B = \begin{pmatrix} 7 & -1 \\ 2 & 3 \\ 0 & 1 \end{pmatrix}$.

Let B^T represent the transpose of B . Compute the inverse of $A + BB^T$, both over the rational numbers and over the integers modulo 7.

[10 Marks]**SOLUTION 4**

First, we consider the case when the matrices are over the rationals

```
export(linalg):
A:=matrix([[1,3,0],[ -1,2,7],[0,8,1]]);
```

```
+ -      - +
|  1, 3, 0  |
|           |
| -1, 2, 7  |
|           |
|  0, 8, 1  |
+ -      - +
```

```
B:=matrix([[7,-1],[2,3],[0,1]]);
```

```
+ -      - +
|  7, -1  |
|         |
|  2,  3  |
|         |
|  0,  1  |
+ -      - +
```

```
BT:=transpose(B);
```

```
+ -      - +
|  7, 2, 0  |
|           |
| -1, 3, 1  |
+ -      - +
```

```
C:=A+B*BT;
```

```
+ -      - +
|  51, 14, -1  |
|           |
|  10, 15, 10  |
|           |
| -1, 11,  2  |
+ -      - +
```

```
CInv:=1/C;
```

```
+ -      - +
|  16/925,  39/4625, -31/925  |
|           |
|  6/925, -101/4625, 104/925  |
|           |
| -1/37,   23/185,   -5/37   |
+ -      - +
```

```
+-                +-

```

Next, we consider the case that the matrices are over the field of integers mod 7

```
MatI7:=Dom::Matrix(Dom::IntegerMod(7));
```

```
Dom::Matrix(Dom::IntegerMod(7))
```

```
A7:=MatI7([[1,3,0],[-1,2,7],[0,8,1]]);
```

```
+-                +-
| 1 mod 7, 3 mod 7, 0 mod 7 |
|                               |
| 6 mod 7, 2 mod 7, 0 mod 7 |
|                               |
| 0 mod 7, 1 mod 7, 1 mod 7 |
+-                +-

```

```
B7:=MatI7([[7,-1],[2,3],[0,1]]);
```

```
+-                +-
| 0 mod 7, 6 mod 7 |
|                               |
| 2 mod 7, 3 mod 7 |
|                               |
| 0 mod 7, 1 mod 7 |
+-                +-

```

```
B7T:=transpose(B7);
```

```
+-                +-
| 0 mod 7, 2 mod 7, 0 mod 7 |
|                               |
| 6 mod 7, 3 mod 7, 1 mod 7 |
+-                +-

```

```
C7:=A7+B7*B7T;
```

```
+-                +-
| 2 mod 7, 0 mod 7, 6 mod 7 |
|                               |
| 3 mod 7, 1 mod 7, 3 mod 7 |
|                               |
| 6 mod 7, 4 mod 7, 2 mod 7 |
+-                +-

```

```
CInv7:=1/C7;
```

```

+-          +-
|  2 mod 7, 5 mod 7, 4 mod 7  |
|                               |
|  6 mod 7, 5 mod 7, 6 mod 7  |
|                               |
|  3 mod 7, 3 mod 7, 1 mod 7  |
+-          +-

```

[10 Marks]

QUESTION 5

Define the function f , from the positive integers to the positive integers

$$f(x) = \begin{cases} 3x + 1 & \text{for odd } x \\ \frac{x}{2} & \text{for even } x. \end{cases}$$

Next, define the sequence $x_{i+1} := f(x_i)$. Write a procedure that, on input of a positive integer n , returns the smallest index i with $x_i = 1$

[15 Marks]

SOLUTION 5

```

f:=proc(x:Type::PosInt)
begin
if x mod 2 = 0 then
return(x/2)
else return(3*x+1)
end_if:
end_proc;

proc f(x) ... end
f(4), f(5);
2, 16

seq:=proc(n:Type::PosInt)
local i;
begin
i:=0:
while n<>1 do
n:=f(n):
i:=i+1:
end_while:
return(i):
end_proc;
proc seq(n) ... end
seq(5);
5
seq(2);

```

1
seq(6);
8

[15 Marks]

QUESTION 6

Solve the following algebraic equations, or systems of algebraic equations, and in each case verify your solution(s) by substitution into the original equations

(a) $x + 2y = 3, y + \frac{1}{x} = 1,$

(b) $\sin \frac{x}{2} = -2 \sin \frac{x}{2} \cos 2x, -2\pi \leq x \leq 2\pi.$

[10 Marks]

SOLUTION 6

(a) `eb1:=x+2*y-3: eb2:= y+(1/x)-1: sb:=solve({eb1=0,eb2=0},{x,y});`

```
{[x = -1, y = 2], [x = 2, y = 1/2]}
for i from 1 to 2 do
res[i]:=op(sb,i):
test[i]:=[subs(eb1,res[i]),subs(eb2,res[i])]:
print(test[i]);
end_for;
[0, 0]
```

[0, 0]

(5 Marks)

(b) `ec:=sin(x/2)+2*sin(x/2)*cos(2*x):`

`sc:=solve(ec=0,x,Domain=Dom::Interval([-2*PI,2*PI]));`

```
{
          PI  PI    2 PI  2 PI    4 PI  4 PI    5 PI  5 PI }
{ 0, -2 PI, 2 PI, - ---, ---, - ----, ----, - ----, ----, - ----, ---- }
{
          3   3     3     3     3     3     3     3   }

```

```
for i from 1 to nops(sc) do
res[i]:=op(sc,i):
test[i]:=subs(ec,x=res[i]):
print(test[i]);
end_for;
0
```

0

0

0
0
0
0
0
0
0
0

(5 Marks)
[10 Marks]

QUESTION 7

Find the first five terms of the asymptotic expansion of $\sqrt{x+1} - \sqrt{x-1}$.

[10 Marks]

SOLUTION 7

f:=sqrt(x+1)-sqrt(x-1);

$$(x + 1)^{1/2} - (x - 1)^{1/2}$$

series(f,x=infinity,12);

$$\frac{1}{x^{1/2}} + \frac{1}{8x^{5/2}} + \frac{7}{128x^{9/2}} + \frac{33}{1024x^{13/2}} + \frac{715}{32768x^{17/2}} + \frac{4199}{262144x^{21/2}} + \dots$$

```

/ 1 \
O| ---- |
| 23/2 |
\ x  /

```

[10 Marks]

QUESTION 8

Use MuPAD to show that

$$\frac{4^{x-1} + 2^{2x-4}}{2^{2x+1} + 5 \times 2^{2x-3}} = \frac{5}{42}$$

[10 Marks]

SOLUTION 8

```
LHS:=(4^(x-1)+2^(2*x-4))/(2^(2*x+1)+5*2^(2*x-3)):
RHS:=5/42:
a4:=simplify(LHS-RHS)
```

$$\frac{4^{x-1} + 2^{2x-4}}{5 \cdot 2^{2x} + 2} - \frac{5}{42}$$

```
a4:=subs(a4,4^(x-1)=2^(2*(x-1))):
a4:=normal(a4);
```

0

[10 Marks]**QUESTION 9**

Solve the following system of equations

1. $x_{i-1} - (2 - h^2)x_i + x_{i+1} = 0$, $i = 2, \dots, 49$, $x_1 = 1$, and $x_{50} - x_{49} = h$, where $h = 0.1$.

[15 Marks]**SOLUTION 9**

```
x[1]:=1:
h:=0.1:
eqs:=[x[i-1]-(2-h^2)*x[i]+x[i+1]]=0 $ i =2..49,
x[50]-x[49]=h]:
vars:=[x[i] $ i =2..50]:
numeric::linsolve(eqs,vars);
```

```
[x[2] = 1.001984326, x[3] = 0.9939488083, x[4] = 0.9759738027,
x[5] = 0.9482390591, x[6] = 0.9110219249, x[7] = 0.8646945715,
x[8] = 0.8097202723, x[9] = 0.7466487704, x[10] = 0.6761107809,
x[11] = 0.5988116835, x[12] = 0.5155244693, x[13] = 0.4270820113,
x[14] = 0.3343687313, x[15] = 0.238311765, x[16] = 0.139871681,
```

x[17] = 0.0400328802, x[18] = -0.06020624939, x[19] = -0.1598433165,
x[20] = -0.2578819504, x[21] = -0.3533417649, x[22] = -0.4452681616,
x[23] = -0.5327418768, x[24] = -0.6148881732, x[25] = -0.6908855879,
x[26] = -0.7599741467, x[27] = -0.821462964, x[28] = -0.8747371517,
x[29] = -0.9192639678, x[30] = -0.9545981443, x[31] = -0.9803863394,
x[32] = -0.996370671, x[33] = -1.002391296, x[34] = -0.9983880079,
x[35] = -0.9844008398, x[36] = -0.9605696633, x[37] = -0.9271327902,
x[38] = -0.8844245892, x[39] = -0.8328721423, x[40] = -0.7729909739,
x[41] = -0.7053798958, x[42] = -0.6307150188, x[43] = -0.5497429915,
x[44] = -0.4632735344, x[45] = -0.3721713419, x[46] = -0.277347436,
x[47] = -0.1797500557, x[48] = -0.08035517489, x[49] = 0.01984325769,
x[50] = 0.1198432577]

[15 Marks]